
ABSTRACT: In this paper we will explore ways in
which complex integrals can be constructed
geometrically using GeoGebra. We will try to
deduce laws of complex integration by experimental
means.

1 Introduction
There exists a common misconception that complex integration is too harsh
to be visualized. Still we can find some geometrical representations [WE12,
LAN99, NEE98, AH10]. As it turns out complex integral can be visualized
as path [LAN99] or be constructed using vectors [AH10], there are also
other ideas of visualizing complex integral [WE12]. In “Visual Complex
Analysis” Tristan Needham did tremendous work on this subject trying to
show that complex integration is just continuation of real integral and can be
produced in a plane using complex numbers, vectors and sums [NEE98]. In
this paper we will follow Tristan Needham's ideas and expand them to suit
GeoGebra (more information about dynamical geometry software
GeoGebra: http://www.geogebra.org).
Riemann integral (usually called real integral) is usually calculated
following idea of Riemann sum or trapezoidal rule, but idea in both cases is
the same - adding up function value at the point in interval multiplied by
size of interval as this mentioned interval length goes to zero [PU10]. This

77

idea geometrically is visualized as area under the curve. Formally, let
a = x0 < x1 < ... < xn = b, then we get

 f xdx = lim  f t x
k

k

, t k  xk 1 , xk .

We will try to extend this idea to complex plane. First we will choose
two points in and then connect them, then chop this path to small intervals
and find function value obtained in this interval. Multiplying obtained
intervals and values and summing them up we should obtain complex
integral value. Formally we have such plan:
1. Create points a   and b   .
2. Create path between a and b , then chop it to small pieces  i .
3. Find points z i inside obtained intervals.
4. Calculate function value at each point f z i  .
5. Calculate sum

 f z 
i

i

.

2 Constructing complex integral
Let us evaluate integral

 e dz . We get into difficulties right away, because
z

in complex plane there are infinite ways we can connect two points. For our
first graph let a path between these two points be a straight line. We are
forced to use vectors to represent this path.
In this paper we will use Sequence command in GeoGebra to create
lists of elements, but Spreadsheet can be used to make same constructions.
All constructions used in this paper and more can be found at:
https://sites.google.com/site/visualcomplexanalysis/.
1. Let a   and b   .
>>> a=0+i
>>> b=1+0*i
2. We create vectors  i . In this case we will create 10 equal vectors
from point a to b.
>>> Sequence[Vector[a+n*Vector[a,b]/10, a+(n+1)*Vector[a,b]/10],
n, 0, 9]
3. Now we need a list of points in the middle of these vectors,
formally called z i .
>>> Sequence[a+n*Vector[a,b]/10+Vector[a,b]/20, n, 0, 9]

4. Now we calculate values at points in list2, thus obtaining

f zi  = e i . Later we will be able to experiment with different functions.
>>> Sequence[e^Element[list2,n], n, 1, 10]
5. We will multiply obtained values by representative vectors
obtaining f z i  i and then add these vectors to get value of our integral,
z

which is equal to

 f z 
i

i

.

>>> Sequence[(Element[list1, n]+0*i) Element[list3, n], n, 1, 10]
>>> Sequence[Sum[Sequence[Element[list4,n],n,1,m]]+0*i,m,1,10]

Fig. 1. Visual representation of  e z dz.
Obtained sum (see Fig. 1) with each step gets closer to point
2.18  0.84i , which is actually approximately equal to actual integral value.
To get more accurate result we can make  i variable and decrease its size
till we get desired accuracy. From real case it is good guess that complex
integrals could obey same integration laws as real ones and  e z dz = e b  e a .
We can visually make sure that this is true by adding a new point:
>>> e^b-e^a
Now by dragging point a to the position of point b and point a to
position of point b we are changing integration order. It is easy to see that in
such case we obtain  2.18+ 0.84i , thus the sign of the function has
changed. Analogous to real case we just witnessed the property of complex
b

∫a

f ( z)dz= − ∫

a

f (z)dz .
b
integration
In the same way it is possible to convince yourself with other complex
integration properties, see [INT84] or any other introductory course book to
complex analysis.

It is easy to change integration path to arc (see Fig. 2) or even random
path (we will leave this as exercise to reader). Still in each case we will get
same result equal to e b  e a . Hence we are forced to come up with idea that
changing integration path will not affect integral value. For formal proof
look at [AH10].

Fig. 2. Visual representation of  e z dz. Arc path.

2 Closed circular path
While calculating real integral choosing start point and end point to be equal
was impossible task. In complex plane we can make same point to be
starting point and end point and define path from start to finish. For
simplicity we will create circular path, as we saw already circular path can
be deformed and this will not affect integral value. We will explore
integral  1 / zdz.
It is easy to create circular path in GeoGebra using polar coordinate
form, all other steps are similar to previous example.
>>> Sequence[Vector[(1;n pi/18),(1;(n+1) pi/18)],n,0,35]
>>> Sequence[(1;n pi/18+pi/36)+0*i,n,0,35]
>>> Sequence[1/Element[list2,n], n, 1, 36]
>>> Sequence[(Element[list1, n]+0*i) Element[list3, n], n, 1, 36]
>>> Sequence[Sum[Sequence[Element[list4,n],n,1,m]]+0*i,m,1,36]

Fig. 3. Visual representation of  1 / zdz.
We just obtained astounding result

1/ zdz = 2πi

(see Fig. 3). As it

turns out it has something to do with z  0 . We will call this point
singularity and explore this phenomenon further.

3 Exploring General residue theorem
We will try to evaluate

 1 /z  2idz .

This function has singularity at

z = 2i , thus again as in previous example we expect something strange to
happen as we encircle this point. In GeoGebra we create construction with
movable integration path to make it easier to see how changing integration
circle position will affect integral value.
>>> b=0+0*i
>>> Sequence[Vector[b+(1;n pi/18),b+(1;(n+1) pi/18)],n,0,35]
>>> Sequence[b+(1;n pi/18+pi/36)+0*i,n,0,35]
>>> Sequence[1/(Element[list2,n]-2*i), n, 1, 36]
>>> Sequence[(Element[list1, n]+0*i) Element[list3, n], n, 1, 36]
>>> Sequence[Sum[Sequence[Element[list4,n],n,1,m]]+0*i,m,1,36]

Fig. 4. Integral  1 / z  2i dz near origin.
We see that near origin integral value goes to zero (see Fig. 4). From
this point we can move point b and see what happens near 2πi (see Fig. 5).
We can try to change GeoGebra construction to encircle singularity
point multiple times and see that if we encircle singularity n times we get
2πin , thus let us denote number of rotations as vK, s j  . Changing function

to f z  = 2 / z  2i  or even to f z  = z / z  2i  will enable us to see that
integral value depends on numerator, we will denote it Res  f z , s j  . From
this point it is easy to come up with General Residue Theorem
 f z dz = 2πi vK,s j Res  f z ,s j . From here reader is invited to read
formal proof [AH10].

4 Exploring Cauchy formula
Following the same construction idea, from previous example, we can
create graphical representation of integral  z 2 / z  a dz only by creating
new point and redefining list3 as follows:
>>> a=1+i
>>> list3=Sequence[Element[list2,n]^2/(Element[list2,n]-a), n, 1, 36]

Fig. 5. Integral  1 / z  2i dz near z = 2i .
It is quite easy to convince yourself with truthfulness of Cauchy
Formula  f z  / z  a dz = 2πif a  from General Residue Theorem, or with
the help of GeoGebra construction we just made (see Fig. 6).

Fig. 6. Integral

 z /z  2idz .

5 Final thoughts
Author hopes that constructions you just made will inspire to explore
complex integration further. You can find a lot of beautiful problems in
Tristan Needham “Visual Complex Analysis” [NEE98]. There are few ideas
what to do next: compute other integrals and construct different integration
paths; explore what happens if you encircle singularity n times; explore
what happens if integral has couple singularities and you encircle just some
of them.
Geometrical reasoning this time was not strict enough to hold it as
proof on its own, it was given following hope that this will enable interested
reader to understand properties of complex integration in more vivid
geometrical manner. It is strongly recommended to read more about
complex integration [NEE98] and then come back to GeoGebra.
More information about visualizing complex variable and constructing
complex integrals with help of Spreadsheet in GeoGebra can be found at
https://sites.google.com/site/visualcomplexanalysis/.

